

# Pengenalan Genius Mathematics Tools

Saifuddin Arief  
[Saifuddin.Arief@rocketmail.com](mailto:Saifuddin.Arief@rocketmail.com)

## Lisensi Dokumen:


Copyright © 2003-2018 IlmuKomputer.Com

Seluruh dokumen di IlmuKomputer.Com dapat digunakan, dimodifikasi dan disebarkan secara bebas untuk tujuan bukan komersial (nonprofit), dengan syarat tidak menghapus atau merubah atribut penulis dan pernyataan copyright yang disertakan dalam setiap dokumen. Tidak diperbolehkan melakukan penulisan ulang, kecuali mendapatkan ijin terlebih dahulu dari IlmuKomputer.Com.

Genius merupakan *freeware* untuk komputasi numerik dan visualisasi data. Genius cocok digunakan sebagai sebuah kalkulator biasa maupun untuk menyelesaikan sejumlah persoalan numerik yang muncul dalam dunia rekayasa dan penelitian. Genius hanya dapat dijalankan pada sistem operasi Linux. Informasi dan dokumentasi dan yang terkait dengan Genius dapat dilihat pada situs <https://www.jirka.org/genius.html#use>.

## Menjalankan program Genius

Genius dapat dijalankan melalui menu utama atau melalui terminal dengan menggunakan perintah `gnome-genius`. Setelah Genius dijalankan maka pada layar komputer akan muncul jendela Genius seperti yang terlihat pada Gambar 1.



Gambar 1. Genius

Simbol `genius>` yang terdapat pada jendela Genius merupakan tanda bahwa Genius siap untuk menerima suatu perintah. Sebagai contoh, untuk melakukan perhitungan  $1.23 + 4.56$  maka ekspresi matematika tersebut diketikkan setelah simbol `>>` kemudian ditekan tombol `Enter` untuk menjalankan perintahnya. Genius akan menampilkan hasil perhitungannya pada baris berikutnya.

```
genius> 1.23 + 4.56
= 5.79
```

Tanda `genius>` yang muncul kembali menunjukkan bahwa Genius siap untuk mengerjakan suatu perintah yang lain.

Melalui menu `Help` atau perintah `manual` dapat diperoleh penjelasan atau manual mengenai penggunaan program Genius. Kemudian untuk keluar dari program Genius dapat dilakukan dengan menggunakan perintah `exit` atau `quit`, melalui menu `File - Quit` atau dengan menekan tombol `×` yang terletak pada bagian kanan atas dari jendela Genius.

## Variabel dan Ekspresi

Variabel adalah sebuah nama yang digunakan untuk menyimpan nilai suatu obyek. Penyimpanan suatu nilai ke dalam suatu variabel dilakukan dengan statemen sebagai berikut:

```
x = ekspresi atau x := ekspresi
```

dimana `x` adalah nama variabel dan `ekspresi` adalah suatu ekspresi matematika. Dalam penggunaan variabel, huruf kecil dan huruf besar adalah berbeda. Nilai dari suatu ekspresi akan ditampilkan pada baris berikutnya kecuali jika ditambahkan tanda titik koma (`;`) pada akhir ekspresinya.

Di dalam Genius terdapat beberapa variabel khusus yang menyatakan suatu konstanta matematika, seperti `pi` untuk  $\pi = 3.1415927\dots$ , dan `e` untuk  $e = 2.7182818\dots$

Untuk memperjelas perintah-perintah yang dibuat, dapat ditambahkan suatu baris komentar. Baris komentar dapat ditulis sebagai suatu baris tersendiri atau ditulis di belakang suatu statemen. Kumpulan karakter yang terletak setelah tanda `#` akan dianggap sebagai baris komentar.

```
genius> lebar = 12.5
= 12.5
```

```
genius> tinggi = 8;
```

```
genius> luas = lebar*tinggi
= 100.0
```

```
genius> r = 10;          # radius lingkaran
```

```
genius> A = pi*r^2      # luas lingkaran
= 314.159265359
```

## Operator-operator dan Fungsi-fungsi Matematika

Operator-operator untuk perhitungan aritmatika adalah sama dengan operator-operator yang terdapat pada kalkulator atau perangkat lunak lainnya yaitu `+`, `-`, `*`, `/` dan `^`. Dimana tanda-tanda tersebut masing-masing adalah simbol untuk operasi penjumlahan, pengurangan,

perkalian, pembagian serta pemangkatan. Di dalam Genius juga telah terpasang fungsi-fungsi yang diperlukan dalam perhitungan matematika, seperti sqrt, abs, exp, sin, cos, tan.

```
genius> 1010*(sqrt(9.81^2 + 7^2)*((0.07 + 0.0214)*cos(35.5/180*pi)))  
= 905.713605859
```

```
genius> p = 101350*exp(-9.807*5000/(287*288.16))  
= 56017.6824426
```

```
genius> k = 0.08567/(pi*(5.8^2 - 4.5^2))*log(25/3.5)  
= 4.00411701102e-3
```

Pada suatu ekspresi matematika, tanda minus yang terdapat didepan suatu bilangan dianggap sebagai menyatu dengan bilangan tersebut bukan sebagai operator minus, seperti yang diilustrasikan pada contoh di bawah ini.

```
genius> -10^2      # akan dievaluasi sebagai (-10)^2  
= 100
```

```
genius> p = 10;
```

```
genius> q = -p^2  
= -100
```

## Bilangan Kompleks

Bilangan kompleks  $z = x + iy$  dinyatakan dengan notasi  $z = x + yi$  dimana  $i$  adalah variabel khusus untuk konstanta  $\sqrt{-1}$ . Operasi aritmatika terhadap bilangan kompleks dapat dilakukan dengan menggunakan notasi yang sama dengan notasi pada bilangan real, seperti pada contoh berikut ini.

```
genius> z2 = 3 + 1i  
= 3+1i
```

```
genius> z1 = 6 - 8i  
= 6-8i
```

```
genius> z2 = 3 + 1i  
= 3+1i
```

```
genius> p = z1 + z2  
= 9-7i
```

```
genius> q = z1 - z2  
= 3-9i
```

```
genius> abs(z1)  
= 10
```

```
genius> z1*z2  
= 26-18i
```

```
genius> z2/z1  
= 1/10+3i/10
```

```
genius> float(ans)  
= 0.1+0.3i
```

## Matrik dan Vektor

Matrik dan vektor dapat dibuat dengan menggunakan operator kurung siku ([ ]). Dimana elemen-elemen matrik atau vektor dimasukkan diantara kedua kurung siku tersebut. Untuk memisahkan elemen yang satu dengan elemen yang lainnya yang terletak pada satu baris dapat digunakan tanda koma (,) dan antara baris yang satu dengan yang lainnya dipisahkan dengan tanda titik koma (;) atau tombol Enter.

```
genius> X = [1,2,3; 4,5,6; 7,8,9]
=
[1, 2, 3
 4, 5, 6
 7, 8, 9]
```

```
genius> X = [1,2,3
             > 4,5,6
             > 7,8,9]
=
[1, 2, 3
 4, 5, 6
 7, 8, 9]
```

```
genius> v = [1,2,3,4,5]
=
[1, 2, 3, 4, 5]
```

```
genius> w = [3;4;1;2]
=
[3
 4
 1
 2]
```

Untuk membuat suatu vektor baris dimana nilai dari elemen-elemennya berubah secara konstan dari suatu nilai awal sampai nilai akhir tertentu dapat digunakan operator tanda titik dua (:). Notasi pembuatan vektor indek adalah i:j:k. Apabila nilai j sama satu maka notasi tersebut dapat ditulis dengan notasi yang lebih singkat yaitu i:k.

```
genius> i = 1:10
=
`[1, 2, 3, 4, 5, 6, 7, 8, 9, 10]
```

```
genius> n = 0:0.25:1
=
`[0, 0.25, 0.5, 0.75, 1.0]
```

```
genius> m = 1:2:10
=
`[1, 3, 5, 7, 9]
```

Genius juga menyediakan sejumlah fungsi yang dapat digunakan untuk membuat matrik-matrik khusus, seperti yang diilustrasikan pada contoh-contoh di bawah ini.

```
genius> A = zeros(3,4)
=
[0, 0, 0, 0
 0, 0, 0, 0
 0, 0, 0, 0]
```

```
genius> B = ones(2,5)
=
[1, 1, 1, 1, 1
 1, 1, 1, 1, 1]
```

```
genius> Y = eye(3)
=
[1, 0, 0
 0, 1, 0
 0, 0, 1]
```

```
genius> D = diag([1,3,5])
=
[1, 0, 0
 0, 3, 0
 0, 0, 5]
```

```
genius> R = rand(3,4)
=
[ 0.332860697499, 0.179963491441, 0.23039838507, 0.149360489174
 0.284576277608, 0.525782954021, 0.175674884675, 0.869105341814
 8.1815294142e-3, 0.217742752611, 0.634600811967, 0.839093286126]
```

Elemen yang terletak pada baris ke-i dan kolom ke-j dari suatu matrik X dapat diekstrak dengan notasi  $X@(i,j)$ .

```
genius> D@(2,2)
= 3
```

```
genius> R@(2,4)
= 0.869105341814
```

## Operasi Berbasis Vektor

Di dalam Genius, secara umum operasi-operasi matematika terhadap obyek matrik dan vektor dapat dilakukan dengan mudah tanpa harus menggunakan suatu perulangan, sebagaimana yang diilustrasikan pada contoh-contoh di bawah ini.

```
genius> x = 0:pi/4:pi
=
`[0, 0.785398163397, 1.57079632679, 2.35619449019, 3.14159265359]
```

```
genius> cos(x)
=
`[1.0, 0.707106781187, 9.4152053883e-40, -0.707106781187, -1.0]
```

```
genius> u = [1,2,3];
```

```
genius> y = e.^u
=
[2.71828182846, 7.38905609893, 20.0855369232]
```

```
genius> log(y)
=
[1.0, 2.0, 3.0]
```

```
genius> round(y)
=
[3, 7, 20]
```

Operasi Aljabar Linier, seperti penjumlahan, pengurangan dan perkalian, juga dapat dilakukan dengan sangat mudah tanpa harus menggunakan suatu ekspresi perulangan secara eksplisit. Pada operasi aljabar argumen-argumennya harus mempunyai dimensi yang kompatibel, jika dimensinya tidak kompatibel maka operasinya tidak dapat dieksekusi dan akan muncul suatu pesan kesalahan.

```
genius> X = [9,8,5;1,3,0; 2,4,6]
=
[9, 8, 5
 1, 3, 0
 2, 4, 6]
```

```
genius> Y = [3,2,1; 4,5,6; 9,8,7]
=
[3, 2, 1
 4, 5, 6
 9, 8, 7]
```

```
genius> A = X + Y
=
[12, 10, 6
 5, 8, 6
 11, 12, 13]
```

```
genius> B = X - Y
=
[ 6, 6, 4
 -3, -2, -6
 -7, -4, -1]
```

```
genius> p =[1,2; 3,8]
=
[1, 2
 3, 8]
```

```
genius> Z = A + p
Can't add/subtract two matrices of different sizes
= ([12,10,6;5,8,6;11,12,13]+[1,2;3,8])
```

```
genius> j = [6;4]
=
[6
 4]
```

```
genius> k = ones(3)
=
[1, 1, 1]
```

```
genius> g = p*j
=
[14
 50]
```

```
genius> h = p*k
Can't multiply matrices of wrong sizes
= ([1,2;3,8]*[1,1,1])
```

Selain operasi perkalian yang mengikuti aturan dalam Aljabar Linier, dapat juga dilakukan operasi terhadap obyek matrik dan vektor dengan melakukan operasi secara elemen dengan

elemen. Operasi elemen dengan elemen juga dapat diterapkan pada operasi pembagian. Notasi untuk perkalian dan pembagian secara elemen dengan elemen adalah  $\cdot$  dan  $\div$ .

```
genius> x =[1,2; 3,4]
=
[1, 2
 3, 4]
```


```
genius> y = [5,6; 7,8]
=
[5, 6
 7, 8]
```

```
genius> u = x.*y
=
[ 5, 12
 21, 32]
```


```
genius> z = y./x
=
[ 5, 3
 2 1/3, 2]
```

## Visualisasi Grafik

Genius dapat digunakan untuk melakukan visualisasi data, baik secara dua dimensi maupun tiga dimensi. Operasi tersebut dapat dilakukan melalui menu Calculator - Plot. Grafik dua dimensi dibuat melalui form Function line plot, kemudian untuk grafik tiga dimensi melalui form Surface plot, seperti yang diperlihatkan pada contoh-contoh di bawah ini.



Gambar 3. Contoh pembuatan grafik dua dimensi melalui form Function line plot




Gambar 4. Ilustrasi visualisasi grafik 2D

The figure shows a "Create Plot" dialog box with two tabs: "Function line plot" (selected) and "Surface plot". The "Function / Expression" section contains a text input field with the expression  $\sin(\sqrt{x^2 + y^2})/\sqrt{x^2 + y^2}$  and a green checkmark icon. Below this is a "Draw legend" checkbox (unchecked) and a "Change variable names..." button. The "Plot Window" section has three input fields: "x from:" and "to:" both set to -10.00000 and 10.00000 respectively; "y from:" and "to:" both set to -10.00000 and 10.00000 respectively; and "Dependent axis from:" and "to:" both set to -10.00000 and 10.00000 respectively. A "Fit dependent axis" checkbox is checked. At the bottom right, there are "Close" and "Plot" buttons.

Gambar 5. Contoh pembuatan grafik dua dimensi melalui form Function line plot





Gambar 6. Ilustrasi visualisasi grafik 3D

## Pemrograman


Genius menyediakan sejumlah kontrol pemrograman yang dapat digunakan untuk mengatur jalannya eksekusi suatu program dengan menggunakan statemen perulangan dan kondisional. Ilustrasi penggunaan statemen perulangan for dan kondisional if diberikan pada contoh-contoh di bawah ini.

```
genius> H = zeros(4,4);
genius> for i = 1 to 4 do (
    >   for j = 1 to 4 do (
    >     H@(i,j) = 1/(i+j-1);)
    > )
genius> H
=
[ 1, 1/2, 1/3, 1/4
  1/2, 1/3, 1/4, 1/5
  1/3, 1/4, 1/5, 1/6
  1/4, 1/5, 1/6, 1/7]

genius> z = zeros(3);
genius> for i = 1 to 3 do (
    >   for j = 1 to 3 do (
    >     if (i==j) then z@(i,i) = i;)
    > )
genius> z
=
[1, 0, 0
 0, 2, 0
 0, 0, 3]
```

## Program

Program adalah sebuah file teks yang di dalamnya terdapat perintah-perintah Genius. Apabila suatu program dijalankan maka perintah-perintah yang terdapat di dalamnya akan dieksekusi seperti seolah-olah kita mengetikkannya pada jendela perintah. Pembuatan suatu program dapat dilakukan melalui menu **File - New Program**. Berikut ini adalah contoh sebuah program (`fibonacci.gel`) yang menggambarkan perhitungan 10 suku pertama dari deret Fibonacci.



```
1 # Deret Fibonacci
2 n = 10;
3 f = zeros(1,n);
4 f@(1) = 1;
5 f@(2) = 2;
6 for k = 3 to n do (
7   f@(k) = f@(k-1) + f@(k-2)
8 );
9 print("10 suku pertama deret Fibonacci")
10 print(f)
11 |
```

Gambar 7. Program `Fibo.gel`

Statemen-statemen yang terdapat di dalam sebuah program yang sedang dibuka pada jendela Editor dapat dijalankan melalui menu **Calculator - Run**. Jika program `fibonacci.gel` dijalankan maka tampilan berikut ini akan muncul pada Jendela Genius.

```
genius>
Output from fibonacci.gel (((
10 suku pertama deret Fibonacci
[1, 2, 3, 5, 8, 13, 21, 34, 55, 89]
))) End
```


## Fungsi

Fungsi merupakan kumpulan dari statemen-statemen Genius yang dapat melakukan suatu komputasi atau perhitungan tertentu. Fungsi dapat dibuat dalam bentuk suatu file atau secara inline pada jendela perintah. Untuk suatu fungsi yang digunakan berulang kali maka biasanya fungsi tersebut dibuat dalam bentuk file fungsi, namun untuk suatu fungsi yang hanya digunakan untuk sementara maka dapat dibuat secara inline pada jendela perintah.

Bentuk umum dari suatu fungsi adalah sebagai berikut:

```
fname(in1,in2,...) = (statemen-statemen fungsi)
```

dimana `fname` adalah nama fungsi yang dibuat dan `in1, in2` adalah argumen-argumen input. Ilustrasi dari sebuah file fungsi diberikan pada gambar 8. Fungsi tersebut [`bisect.m`] adalah fungsi untuk mencari akar dari suatu persamaan dengan metode *bisection*.



```
Genius 1.0.24
File Edit Calculator Examples Programs Settings Help
Console bisect.gel x
1 function bisect(f,x1,x2,tol) = (
2 # Penentuan akar persamaan nonlinier dengan metode Bisection
3
4 if (f(x1)*f(x2) > 0) then
5   (print("Akar tidak terdapat di dalam selang [x1,x2]"); return null);
6
7 n = ceil(log(abs(x2 - x1)/tol)/log(2.0)); # jumlah iterasi maksimum
8
9 for i = 1 to n do (
10  x3 = 0.5*(x1 + x2);
11  f3 = f(x3);
12  if (f3 == 0.0) then
13    (return x3);
14  if (f(x2)*f(x3) < 0.0) then
15    (x1 = x3; f1 = f3)
16  else
17    (x2 = x3; f2 = f3);
18 );
19 x = (x1 + x2)/2;
20 return x;
21 );
22 |
Line: 22
```

Gambar 8.

Contoh pembuatan suatu fungsi secara inline adalah sebagai berikut:

```
genius> function fx(x) = (x - exp(-x));
```

Fungsi-fungsi yang baru dibuat dapat digunakan sebagaimana fungsi-fungsi yang telah terpasang pada Genius. Untuk fungsi yang dibuat secara inline dapat langsung digunakan namun untuk fungsi yang dibuat dalam bentuk file harus dipanggil terlebih dahulu dengan perintah File - Load and Run.

Berikut ini adalah contoh penggunaan fungsi bisect dan fx yang telah dibuat.

```
genius> fx(1)
= 0.632120558829
```

```
genius>
Output from /home/ariefs/Documents/prog1.gel (((
))) End
```

```
genius> xsol = bisect(fx,0,1,1e-6)
= 0.567142963409
```

## Komputasi Numerik

Di dalam Genius, berbagai macam persoalan komputasi numerik dapat diselesaikan dengan mudah, seperti yang ditunjukkan pada contoh-contoh di bawah ini.

Sistem persamaan linear  $Ax = b$  dapat diselesaikan dengan mudah menggunakan operator pembagian kiri ( $\backslash$ ). Notasi  $A \backslash b$  adalah ekuivalen dengan  $\text{inv}(A) * b$ .

```
genius> A = [1,2,1,4; 2,0,4,3; 4,2,2,1; -3,1,3,2]
=
[ 1, 2, 1, 4
  2, 0, 4, 3
  4, 2, 2, 1
 -3, 1, 3, 2]
```

```
genius> b = [13; 28; 20; 6]
=
[13
 28
 20
  6]
```

```
genius> x=SolveLinearSystem(A,b)
=
[ 3
 -1
  4
  2]
```

```
genius> A*x
=
[13
 28
 20
  6]
```

Nilai determinan dari matrik A dapat dihitung dengan menggunakan fungsi  $\text{det}(A)$ .

```
genius> det(A)
= -180
```

Inversi dari suatu matrik A dapat dihitung dengan menggunakan notasi  $A^{(-1)}$ .

```
genius> A^(-1)
=
[ 0, 1/12, 1/12, -1/6
 1/15, -31/90, 29/90, 2/9
 -1/5, 7/60, 7/60, 1/6
 4/15, 11/90, -19/90, -1/9]
```

```
genius> A^(-1)*A
=
[1, 0, 0, 0
 0, 1, 0, 0
 0, 0, 1, 0
 0, 0, 0, 1]
```

Penyelesaian suatu persamaan nonlinier  $f(x) = 0$ , dapat dilakukan dengan beberapa fungsi salah satunya adalah `FindRootSecant`. Berikut ini adalah contoh perhitungan akar dari fungsi  $f(x) = x - \exp(-x)$ .

```
genius> function fx(x) = (x - exp(-x));
```

```
genius> FindRootSecant(fx,0,1,1e-6,20)
=
[true, 0.56714329041, 6.0]
```

Diperoleh bahwa akar dari  $f(x) = x - \exp(-x) = 0$  adalah  $x = 0.5671$ .

Terdapat banyak persamaan integral tertentu yang sulit atau bahkan tidak dapat diselesaikan secara analitis, salah satunya adalah seperti yang terdapat pada contoh berikut ini.

$$Q = \int_0^5 \frac{\sin(3x)}{\sqrt{x^2 + x + 1}} dx$$

Di dalam Genius, nilai suatu integral tertentu dapat dihitung dengan menggunakan fungsi `NumericalIntegral`. Misalkan  $h(x)$  adalah fungsi yang diintegrasikan pada persamaan di atas:

```
genius> function h(x) = (sin(3*x)/sqrt(x^2+x+1))
= (`(x)=(sin((3.0*x))/sqrt(((x^2.0)+x)+1.0)))
```

Selanjutnya, nilai integral dapat dihitung dengan menggunakan fungsi `NumericalIntegral` sebagai berikut.

```
genius> Q = NumericalIntegral(hx,0,5)
= 0.364872811394
```

Jawaban yang diperoleh yaitu  $Q = 0.36487$ .